
1 应用程序的内存组成

对于Windows 32来说，系统会给每个进程4GB的地址空间，低端2GB（$00000000-$7FFFFFFF）给用户支配；高端2GB（$80000000-$FFFFFFFF）留给系统使用，这个4G的地址空间叫“虚拟地址表”，虚拟地址表不是真实的内存。

这个“虚拟地址表”上有4GB的空间，每个程序都有这样的一个表，但他们并不会冲突，这样就阻断了一个进程对另一个进程的访问，系统在需要的时候会把他们映射成具体的真实内存地址。

可以使用GlobalMemoryStatus来获取内存信息，获取后的信息放在TMemoryStatus结构体中。

_MEMORYSTATUS = record
  dwLength: DWORD;        {结构长度}
  dwMemoryLoad: DWORD;   {表示可用内存比例的一个整数, 100 表示内存都可用}
  dwTotalPhys: DWORD;      {物理内存总数}
  dwAvailPhys: DWORD;      {可用物理内存总数}
  dwTotalPageFile: DWORD;   {虚拟内存总数}
  dwAvailPageFile: DWORD;   {可用虚拟内存总数}
  dwTotalVirtual: DWORD;     {虚地址表中的地址总数}
  dwAvailVirtual: DWORD;     {虚地址表中可用的地址总数}
end;
TMemoryStatus = _MEMORYSTATUS;

在用户的2GB地址空间中，低端的0-$FFFF是用于空指针分配，高端的$7FFF0000-$7FFFFFFF用于进程的临界区，这两个地址都是禁止访问的。进程真正的私有空间地址是：$10000-$FFEFFFF。我们可以通过GetSystemInfo函数得到证实，函数返回TSystemInfo结构，结构中的lpMinimumApplicationAddress和lpMaximumApplicationAddress分别标识程序可以访问的最低与最高地址，通过GetSystemInfo还能得到一个内存相关的重要参数：页大小（PageSize），通常大小是4K，我们需要知道的是，用VirtualAlloc函数分配的内存就是PageSize（4K）为最小单位，加入我们用VirtualAlloc给一个整数分配内存，就会浪费4092个字节。

2 栈和堆

栈是编译器自动分配释放，存放函数的参数值，局部变量的值，存取偏移是4字节，不会根据需要动态增长，超出范围会报异常；堆是由程序员分配释放，编译器都会默认建一个“堆”，建立“堆”时会同时提交真实内存，这在申请大内存时会很慢，所以默认“堆”也只有1M，但是“堆”没有限制大小，会根据需要动态增长，注意它与数据结构中的堆是两回事，它的分配方式类似于链表，访问“堆”的内容的时候需要先找到这个“堆”，然后再遍历链表，因此“堆”访问会比“栈”慢。

2.1 栈和堆的内存分配比较

栈：有编译器自动分配释放，存放函数的参数值，局部变量的值等，存取偏移是4字节，申请的变量不会初始化，是一个垃圾值。

堆：有程序员分配释放，若不释放，程序结束时由系统回收，分配的内存不会初始化，是一个垃圾值。

2.2 栈和堆的申请方式方面

栈：由系统自动分配，如在函数申明一个局部变量i: Integer；编译器会自动在栈中分配内存。

堆：需要程序员自己申请，并指明大小。如在函数中申请一个结构体New(A)，A指向的内容是在堆中的，但是A的地址却是在栈中的。

2.3 栈和堆的系统响应方面

栈：只要栈的剩余空间大于所申请空间，系统将为程序提供内存，否则将报异常提示栈溢出。 
堆：操作系统有一个记录空闲内存地址的链表，当系统收到程序的申请时，会遍历该链表，寻找第一个空间大于所申请空间的堆结点，然后将该结点从空闲结点链表中删除，并将该结点的空间分配给程序，另外，对于大多数系统，会在这块内存空间中的首地址处记录本次分配的大小，这样代码中的FreeMem语句才能正确的释放本内存空间。另外由于找到的堆结点的大小不一定正好等于申请的大小，系统会自动的将多余的那部分重新放入空闲链表中。 在堆中分配内存的时候会用HeapLock和HeapUnlock加锁，因此在多线程中分配内存是线性的，效率低下。

2.4 栈和堆的大小限制方面

栈：在Windows下, 栈是向低地址扩展的数据结构，是一块连续的内存的区域。这句话的意思是栈顶的地址和栈的最大容量是系统预先规定好的，在WINDOWS下，栈的大小是固定的（是一个编译时就确定的常数），如果申请的空间超过栈的剩余空间时，将提示overflow。因此，能从栈获得的空间较小。 

堆：是向高地址扩展的数据结构，是不连续的内存区域。这是由于系统是用链表来存储的空闲内存地址的，自然是不连续的，而链表的遍历方向是由低地址向高地址。堆的大小受限于计算机系统中有效的虚拟内存。由此可见，堆获得的空间比较灵活，也比较大。

Delphi改变程序栈大小设置是：Project->Options->Linker->Min stack size/Max stack size。

2.5 获取栈的首地址代码

procedure GetStackAddress(var AStackTop, AStackBottom: Cardinal);

begin

  asm

    mov eax, FS:[4];

    sub eax, 3;

    mov AStackTop, eax;

    mov AStackBottom, ebp;

  end;

end;
3 DELPHI的内存实现

DELPHI是在Windows内存管理的基础上，通过GetMem.inc来实现自己的内存管理器，它封装了操作系统API，使得用户可以直接分配内存，而无须考虑内存具体在哪个虚地址空间。这样的好处就是对于开发者来说：内存分配是透明，分配到的任何一块（既定长度的）内存都是连续的，可以通过字节序遍历一个数据结构，或者在长度边界相同的情况下进行强制转换。DELPHI的内存管理为了考虑多线程环境，使用了临界区变量heapLock，因此在多线程环境中，DELPHI缺省的内存管理器性能会相对较差一些。

4 变量初始化

DELPHI默认初始的变量是全局变量和类变量，初始化的规则，是内存块，内存内容全部初始化为#0，指针初始化为nil，别的（包括函数体内变量）都需要手动初始化。

5 用const来提高应用程序在多核多线程下的性能

我们经常在DELPHI中用const来定义常量，用const来保护函数参数，其实在用const保护函数参数还有另一个更为重要的作用，提高应用程序的执行效率，尤其是在多线程多核下效果更明显。原因是：普通的函数参数如Add(AValue: string)，编译器在传入参数的时候先把变量复制一份，然后当成AValue传入Add，函数结束的时候进行销毁，你在参数上加了const，编译器在传入参数的时候不会进行复制，而是直接传地址，并在编译期间检查不能修改AValue值，我们知道DELPHI的内存管理在申请内存的时候是会加锁的，因此如果调用函数频繁，而且没有加const，这样会造成线程排队等候，性能会不如单线程，const只是对string、结构体等非基本类型有提高效率的作用，对Integer等基本类型（栈变量）不起作用。

6 函数返回值

Delphi函数在返回值的处理上，对于32简单数据类型（不包括浮点数类型），是使用eax返回的，如Integer等；对于64位简单数据类型，是使用edx:eax返回的，如int64类型，而对于结构体返回值来说，长度在32位及以下是通过eax返回的，32位以上返回有点区别，不是“32位以上用寄存器返回指针”，而是在调用前传递了一个默认指针参数。如下面例子的function IncX: TTest;函数，在TTest类型长度大于32位时，实际上等同于于procedure IncX(var v: TTest)或者procedure IncX(v: ^TTest)。下面例子非常有意思：

下面的代码中，定义了一个记录类型，定义了2个同名的方法（同不同名无所谓），v1和v2的x字段都初始化为10，分别调用2个方法后。提问： 

1、v1.x与v2.x的值分别为多少？（要求不要在机上测试） 

2、假如从记录定义中去掉y字段，v1.x与v2.x的值又分别为多少？（要求不要在机上测试）

type

TTest = record

    x: Integer;

    y: Integer;

end;

procedure IncX(var v: TTest); overload;

begin

Inc(v.x);

end;

function IncX: TTest; overload;

begin

Inc(Result.x);

end;

var

v1, v2: TTest;


begin

v1.x := 10;

v2.x := 10;

v1 := IncX;

IncX(v2);

end;

试题的结果是：两个都是11

DELPHI的函数返回值一定要初始化，如果没有初始化，则返回是一个随机值（没有初始化返回值，DELPHI编译器会告警）。

7 内存申请和释放

7.1 内存分配常见函数

GetMem和FreeMem、GetMemory和FreeMemory、New和Dispose、StrAlloc和StrDispose、AllocMem、SysGetMem和SysFreeMem。

7.2 GetMem和FreeMem、GetMemory和FreeMemory

由于DELPHI的内存管理都知道分配内存的大小，因此在释放内存的时候，只要给指针地址不用给出长度就可以了。另外提倡用GetMemory和FreeMemory来代替GetMem和FreeMem，因为FreeMemory会判断指针是否为空。

7.3 New和Dispose

New和Dispose是用来管理变体类型内存分配，如变体结构体：

TRecord = record

  Text: string;

  Value: Integer;

end;

PRecord = ^TRecord;

如果用GetMem和FreeMem、GetMemory和FreeMemory来释放，会造成Text的内存没有释放，造成内存泄漏。如果用Dispose来释放指针，要加上定义信息，否则造成内存泄漏，正确写法Dispose(PRecord(Point))。

7.4 StrAlloc和StrDispose

这个函数也是一对，他们分配PChar加一个Cardinal长度，因此一定要用StrDispose释放，否则容易造成4字节的内存泄漏。StrAlloc分配的指针可以使用StrBufSize来获得大小。

7.5 AllocMem

AllocMem是调用GetMem来分配内存，但是它会把内存全部初始化为#0，因此推荐AllocMem代替GetMem和GetMemory。以下写法都是错误的，都会造成字符串没有结尾符。

function SystemPath: string;

begin

  SetLength(Result, GetSystemDirectory(nil, 0));

  GetSystemDirectory(PChar(Result), Length(Result));

  Result := PChar(Result);

end;
function SystemPath: string;

begin

  SetLength(Result, GetSystemDirectory(nil, 0));

  GetSystemDirectory(PChar(Result), Length(Result));

  Result := PChar(Result);

end;
var

  SystemPath: PChar;

  Len: Cardinal;

begin

  Len := GetSystemDirectory(nil, 0);

  SystemPath := GetMemory(Len);

  GetSystemDirectory(SystemPath, Len)

  FreeMem(SystemPath);

end;
SysGetMem和SysFreeMem

SysGetMem和SysFreeMem是上面函数的底层实现，申请的内存不通过DELPHI内存管理器管理，一般不直接使用它们。

8 String和Pchar

8.1 String的结构

String结构为
	Cardinal（引用计数）
	Cardinal（长度）
	内容


8.2 直接给常量赋值，系统会自动分配内存，自动释放

var

  p: PChar;

begin

  p := '常量赋值会自动分配内存';

  ShowMessage(p); 

end;
9 数组和动态数组

9.1 数组到底是在栈中还是在堆中

数组单一说在栈中还是在堆中都是错误，分为几种情况：

9.1.1 固定数组在函数体内分配是在栈中的

我们做一个试验，一般DELPHI程序线程的栈大小是1M，如果我们函数体中申请4M大小的数组，报栈溢出，则表示数据的大小是在栈中的。
const

CArrayCount = 1024*1024*4;

procedure TForm1.btnMainThreadClick(Sender: TObject);

var

MainThreadArray: array[1..CArrayCount] of Char;

i: Integer;

begin

for i := Low(MainThreadArray) to High(MainThreadArray) do

    MainThreadArray[i] := #0;

end;

我把以上代码在主线程中测试时，确实报了栈溢出，如果这时你把DELPHI程序的栈调大为6M则正确，表示在函数体中申请的数组是在栈中的。

9.1.2 固定数组在类中分配是在堆中的

我们在类中加一下定义语句

FFixArray: array[1..CArrayCount] of Char;

程序正常，表示在类中分配固定数组是在堆中的。

9.1.3 固定数组全局变量是在堆中的

我们在程序定义全部数组

GFixArray: array[1..CArrayCount] of Char;

程序也正常，表示全局固定长度是在堆中的。

9.1.4 动态数组不管在函数体中、类中、全局变量都是在堆中的

如果你会汇编，看一下汇编就明白了。DELPHI这么实现是合理的，在函数里中分配的固定长度数组放在栈中可以加快运行效率，而且在多线程的情况下，不用像堆分配有加锁。只是大家在写程序的过程中注意在函数里定义太长的数组需要注意，否则栈溢出，程序就崩溃了。

9.2 DELPHI的数组和动态数组都是自动管理的，不需要手动释放

动态数组用SetLength来分配内容，手动释放可以直接把动态数组长度设为0。

10 结构体

10.1 结构体是在栈中还是堆中

10.1.1 在函数体中定义的结构体是在栈中的

以下代码会直接报堆栈溢出

const

  CMaxStackSize = 4*1024*1024;

type

  TStackRecord = record

    Value: Integer;

    ValueInteger: array[0..CMaxStackSize] of Integer;

  end;

  PStackRecord = ^TStackRecord;

var

  Form1: TForm1;

implementation

{$R *.dfm}

procedure TForm1.FormCreate(Sender: TObject);

var

  StackRecord: TStackRecord;

begin

  ShowMessage('此处会报栈溢出' + IntToStr(Integer(@StackRecord)));

end;
10.1.2 结构体在类中、全局变量都是在堆中

10.1.3 结构体指针，指针地址在栈中，结构体内容在堆中

以下代码不会报堆栈溢出

const

  CMaxStackSize = 4*1024*1024;

type

  TStackRecord = record

    Value: Integer;

    ValueInteger: array[0..CMaxStackSize] of Integer;

  end;

  PStackRecord = ^TStackRecord;

var

  Form1: TForm1;

implementation

{$R *.dfm}

procedure TForm1.FormCreate(Sender: TObject);

var

  StackRecord: PStackRecord;

begin

  New(StackRecord);

  ShowMessage('此处不会报栈溢出' + IntToStr(Integer(@StackRecord)));

end;
10.2 结构体的申请和释放

结构体申请和释放建议使用New和Dispose，不提倡使用GetMem和FreeMem，这是因为如果结果体中存在String和动态数组，用GetMem和FreeMem会造成内存泄漏。

11 内存操作：复制、清空、填充

11.1 内存复制

内存复制是用以下三个函数：MoveMemory、CopyMemory、Move，其中MoveMemory和CopyMemory是同样功能的函数，都是调用Move函数操作，这个我们可以通过查看它们的代码看到：

procedure MoveMemory(Destination: Pointer; Source: Pointer; Length: DWORD);

begin

  Move(Source^, Destination^, Length);

end;

procedure CopyMemory(Destination: Pointer; Source: Pointer; Length: DWORD);

begin

  Move(Source^, Destination^, Length);

end;
Move函数会自动寻址，因此我们调用的时候不能直接给地址，如复制两个整数，正确写法是：

var

  AValue1, AValue2: Integer;

begin

  Move(AValue1, AValue2, SizeOf(Integer));

end;
在DELPHI中，Move函数是汇编实现的，效率非常高，为了方便大家查看我把下面的代码加了注释，图片效果更好看一点。

procedure Move ( const Source; var Dest; count : Integer );

{$IFDEF PUREPASCAL}

var

  S, D: PChar;

  I: Integer;

begin

  S := PChar(@Source);

  D := PChar(@Dest);

  if S = D then Exit;

  if Cardinal(D) > Cardinal(S) then

    for I := count-1 downto 0 do

      D[I] := S[I]

  else

    for I := 0 to count-1 do

      D[I] := S[I];

end;

{$ELSE}

asm

{     ->EAX     Pointer to source       }

{       EDX     Pointer to destination  }

{       ECX     Count                   }

        PUSH    ESI     //在DELPHI的内嵌汇编中，只有EAX、EDX、ECX三个寄存器可以自由使用

        PUSH    EDI     //别的都需要在使用之后还原，一般采用的是压栈出栈方式保护

        MOV     ESI,EAX //EAX接收第一个参数，因此EAX指向Source，ESI = Source

        MOV     EDI,EDX //EDX接收第二个参数，因此EDX指向DEST，EDI = Dest

        MOV     EAX,ECX //保存ECX到EAX中，利于后面还原

        CMP     EDI,ESI //比较Source和Dest

        JA      @@down  //DEST高于Source，则跳转到down

        JE      @@exit  //内存地址相同则退出

        SAR     ECX,2   //ECX := ECX DIV 4

        JS      @@exit  //负号跳转

        REP     MOVSD   //按每次4字节进行循环拷贝

        MOV     ECX,EAX //还原ECX

        AND     ECX,03H //ECX := ECX mod 4

        REP     MOVSB   //按单字节拷贝循环拷贝剩余字节

        JMP     @@exit  //退出

@@down:

        LEA     ESI,[ESI+ECX-4] //ESI指向Source的最后不足4字节的地址

        LEA     EDI,[EDI+ECX-4] //EDI指向Source的最后不足4字节的地址

        SAR     ECX,2           //ECX := ECX DIV 4

        JS      @@exit          //负号跳转

        STD                     //方向设置

        REP     MOVSD           //按每次4字节进行循环拷贝

        MOV     ECX,EAX         //还原ECX

        AND     ECX,03H         //ECX := ECX mod 4

        ADD     ESI,4-1         //不足4字节的地址

        ADD     EDI,4-1         //不足4字节的地址

        REP     MOVSB           //按单字节拷贝循环拷贝剩余字节

        CLD

@@exit:

        POP     EDI     //还原EDI

        POP     ESI     //还原ESI

end;

{$ENDIF}
我们知道32位机器按4字节拷贝是最快的，从上面的代码可以看出，它采用也是这种做法，先循环按4字节拷贝，然后循环按1字节拷贝剩下的。

11.2 内存清空与填充

内容清空与填充是用到以下三个函数：FillMemory、ZeroMemory、FillChar，其中FillMemory和ZeroMemory都是调用FillChar函数，这个我们可以从它们的源代码中得出。

procedure FillMemory(Destination: Pointer; Length: DWORD; Fill: Byte);

begin

  FillChar(Destination^, Length, Fill);

end;

procedure ZeroMemory(Destination: Pointer; Length: DWORD);

begin

  FillChar(Destination^, Length, 0);

end;
DELPHI中的FillChar也是用汇编代码实现的，其实现的思路和Move函数相同，都是按4字节赋值。

12 内存映射

内存映射是Windows提供可以在进程之间共享内存的方法，主要是由以下四个函数：CreateFileMapping、OpenFileMapping、MapViewOfFile、UnmapViewOfFile组成，这方面的资料非常多，很多DELPHI的书籍都有介绍。我们这给出一个在系统服务和IIS之间（不通系统账户、不通权限）实现共享内存的例子。

13 FastMM内存管理器

FastMM是一个由法国开发者发起的开源DELPHI内存管理器，它针对Borland内存管理器的很多弊端进行改善，采用纯汇编编写，效率比Borlan内存管理器提高2－4倍，在多线程下表现会更好。从DELPHI 2006开始，DELPHI都采用FastMM来提高其IDE的速度。

13.1 用FastMM加快DELPHI IDE的速度

找到里面的FastMm\Replacement BorlndMM DLL\Precompiled\for Delphi IDE\Performance目录下的BorlndMM.dll替换DELPHI的Bin目录下BorlandMM.dll即可。

13.2 用FastMM加快我们自己应用程序的速度

找到里面的FastMM\Replacement BorlndMM DLL\Delphi\Precompiled\for Applications\Performance目录下的BorlandMM.dll放到我们应用程序目录下即可。如果你的程序只是一个单独的EXE，则在工程的第一句加上FastMM4即可。

13.3 用FastMM检测内存泄漏

用FastMM检测内存泄漏，需要在工程的第一个文件引用FastMM4和FastMM4Messages，在FastMM4Options.inc打开FullDebugMode模式，并把FastMM\FullDebugMode DLL\Precompiled\FastMM_FullDebugMode.dll拷贝应用程序目录下，重新编译程序，这样如果程序有内存泄漏，在退出程序的时候，FastMM会生成一个内存泄漏文档，里面有详细的堆栈信息，非常利于查找。

下面一个内存泄漏的样例，如果要能打出MAP地址的函数，需要在DELPHI中打开生成MAP文件的选项（Project->Project Options->Linker->Map file->Detailed），如果是在发布的电脑上，需要把MAP和EXE放在一起。从FastMM4.84之后，FastMM4Options.inc有中文版，在路径下FastMM\Translations\Chinese (Simplified)，对于理解每个编译选项非常有帮助。

14 常见内存错误、代码审查

14.1 PCHAR指针末尾没有赋#0

14.2 没初始化内存出现错误，用FillChar、ZeroMemory

14.3 函数返回值没有初始化

14.4 PwideChar(string)内存格式不一致，强制转换，导致内存错乱

14.5 申请的函数和释放的函数不一致

15 附录：内存函数

System.AllocMem;

System.AllocMemCount;

System.AllocMemSize;

System.Dispose;

System.Finalize;

System.FreeMem;

System.FreeMemory;

System.GetHeapStatus;

System.GetMem;

System.GetMemory;

System.GetMemoryManager;

System.GetMemoryManagerState;

System.GetMemoryMap;

System.IsMemoryManagerSet;

System.Move;

System.New;

System.ReallocMem;

System.ReallocMemory;

System.SetMemoryManager;

System.SysAllocMem;

System.SysFreeMem;

System.SysGetMem;

System.SysReallocMem;

SysUtils.DisposeStr;

SysUtils.NewStr;

TlHelp32.Heap32First;

TlHelp32.Heap32ListFirst;

TlHelp32.Heap32ListNext;

TlHelp32.Heap32Next;

Windows.CopyMemory;

Windows.FillMemory;

Windows.GetProcessHeap;

Windows.GetProcessHeaps;

Windows.GlobalAlloc;

Windows.GlobalAllocPtr;

Windows.GlobalCompact;

Windows.GlobalDiscard;

Windows.GlobalFix;

Windows.GlobalFlags;

Windows.GlobalFree;

Windows.GlobalFreePtr;

Windows.GlobalHandle;

Windows.GlobalLock;

Windows.GlobalMemoryStatus;

Windows.GlobalReAlloc;

Windows.GlobalReAllocPtr;

Windows.GlobalSize;

Windows.GlobalUnfix;

Windows.GlobalUnlock;

Windows.GlobalUnWire;

Windows.GlobalWire;

Windows.HeapAlloc;

Windows.HeapCompact;

Windows.HeapCreate;

Windows.HeapDestroy;

Windows.HeapFree;

Windows.HeapLock;

Windows.HeapReAlloc;

Windows.HeapSize;

Windows.HeapUnlock;

Windows.HeapValidate;

Windows.HeapWalk;

Windows.IsBadCodePtr;

Windows.IsBadReadPtr;

Windows.IsBadStringPtr;

Windows.IsBadWritePtr;

Windows.MoveMemory;

Windows.VirtualAlloc;

Windows.VirtualAllocEx;

Windows.VirtualFree;

Windows.VirtualFreeEx;

Windows.VirtualLock;

Windows.VirtualProtect;

Windows.VirtualProtectEx;

Windows.VirtualQuery;

Windows.VirtualQueryEx;

Windows.ZeroMemory;

Windows.LocalAlloc;

Windows.LocalCompact;

Windows.LocalDiscard;

Windows.LocalFlags;

Windows.LocalFree;

Windows.LocalLock;

Windows.LocalReAlloc;

Windows.LocalShrink;

Windows.LocalSize;

Windows.LocalUnlock;

_1337087483/DiffUser_DiffSecurity_SharedMemorty.pas
unit DiffUser_DiffSecurity_SharedMemorty;



interface



uses

  Windows, Messages, SysUtils, Math;



const

  CSharedMemSize = 1024*1024*3;

  CSharedMemName = 'Global\DiffUser_DiffSecurity_SharedMemorty';



type

  //报表执行状态

  TReportState = (rsFinished, rsInexistent, rsWaiting, rsRunning, rsException, rsCanceled);

  TShareMem = record

    Handle: THandle;

    ReportID: Integer;

    ReportState: TReportState;

    ReportType: Integer;

    ReportData: array[0..1024*4] of Char;

    DestFile: array[0..1024*4] of Char;

    HintMsg: array[0..1024*3] of Char;

  end;

  PShareMem = ^TShareMem;



  TPublicVars = class

  private

    FShareMem: PShareMem;

    FMapFile: THandle;

    procedure SetHandle(AValue: THandle);

    function GetHandle: THandle;

    procedure SetReportID(AValue: Integer);

    function GetReportID: Integer;

    procedure SetReportState(AValue: TReportState);

    function GetReportState: TReportState;

    procedure SetReportType(AValue: Integer);

    function GetReportType: Integer;

    procedure SetReportData(AValue: string);

    function GetReportData: string;

    procedure SetDestFile(AValue: string);

    function GetDestFile: string;

    procedure SetHintMsg(AValue: string);

    function GetHintMsg: string;

  public

    constructor Create(ANew: Boolean);

    destructor Destroy; override;

    property Memory: PShareMem read FShareMem;

    property Handle: THandle read GetHandle write SetHandle;

    property ReportID: Integer read GetReportID write SetReportID;

    property ReportState: TReportState read GetReportState write SetReportState;

    property ReportType: Integer read GetReportType write SetReportType;

    property ReportData: string read GetReportData write SetReportData;

    property DestFile: string read GetDestFile write SetDestFile;

    property HintMsg: string read GetHintMsg write SetHintMsg;

  end;



implementation



{ TPublicVars }

resourcestring

  CouldNotMapViewOfFile = 'Could not map view of file.';



const

  SECURITY_NULL_SID_AUTHORITY = 0;

  SECURITY_WORLD_SID_AUTHORITY = 1;

  SECURITY_LOCAL_SID_AUTHORITY = 2;

  SECURITY_CREATOR_SID_AUTHORITY = 3;

  SECURITY_NT_AUTHORITY = 5;



  SECURITY_NULL_RID = 0;

  SECURITY_WORLD_RID = 0;

  SECURITY_LOCAL_RID = 0;

  SECURITY_CREATOR_OWNER_RID = 0;

  SECURITY_CREATOR_GROUP_RID = 1;



  ACL_REVISION = 2; 



type

  ACL_SIZE_INFORMATION = record

    AceCount: DWORD;

    AclBytesInUse: DWORD;

    AclBytesFree: DWORD;

  end;



  ACE_HEADER = record

    AceType: BYTE;

    AceFlags: BYTE;

    AceSize: WORD;

  end;

  PACE_HEADER = ^ACE_HEADER;



  ACCESS_ALLOWED_ACE = record

    Header: ACE_HEADER;

    Mask: ACCESS_MASK;

    SidStart: DWORD;

  end;          



constructor TPublicVars.Create(ANew: Boolean);

var

  SecMem: SECURITY_ATTRIBUTES;

  aSD: SECURITY_DESCRIPTOR;

begin

  inherited Create;

  { 创建一个任何用户都可以访问的内核对象访问权 }

  InitializeSecurityDescriptor(@aSD, SECURITY_DESCRIPTOR_REVISION);

  SetSecurityDescriptorDacl(@aSD, True, nil, False);

  SecMem.nLength := SizeOf(SECURITY_ATTRIBUTES);

  SecMem.lpSecurityDescriptor := @aSD;

  SecMem.bInheritHandle := False;

  FMapFile := CreateFileMapping($FFFFFFFF, @SecMem, PAGE_READWRITE, 0, CSharedMemSize, CSharedMemName);

  FMapFile := OpenFileMapping(File_Map_All_Access, False, CSharedMemName);

  if (FMapFile = 0) then

  begin

    raise Exception.Create(SysErrorMessage(GetLastError));

    OutputDebugString(PChar(SysErrorMessage(GetLastError)));

  end

  else

  begin                                   // 成功

    FShareMem := MapViewOfFile(FMapFile, File_Map_All_Access, 0, 0, CSharedMemSize);

    OutputDebugString(PChar(SysErrorMessage(GetLastError) + ',Handle=' + IntToStr(Handle)));

  end;

end;



destructor TPublicVars.Destroy;

begin

  UnmapViewOfFile(FShareMem);

  CloseHandle(FMapFile);

  inherited;

end;



function TPublicVars.GetHandle: THandle;

begin

  Result := FShareMem.Handle;

end;



procedure TPublicVars.SetHandle(AValue: THandle);

begin

  FShareMem.Handle := AValue;

end;



function TPublicVars.GetReportID: Integer;

begin

  Result := FShareMem.ReportID;

end;



procedure TPublicVars.SetReportID(AValue: Integer);

begin

  FShareMem.ReportID := AValue;

end;



function TPublicVars.GetReportType: Integer;

begin

  Result := FShareMem.ReportType;

end;



procedure TPublicVars.SetReportType(AValue: Integer);

begin

  FShareMem.ReportType := AValue;

end;



function TPublicVars.GetReportData: string;

begin

  Result := FShareMem.ReportData;

end;



procedure TPublicVars.SetReportData(AValue: string);

var

  iMaxCount: Integer;

begin

  ZeroMemory(@FShareMem.ReportData[0], Length(FShareMem.ReportData));

  iMaxCount := Min(Length(FShareMem.ReportData), Length(AValue));

  StrMove(@FShareMem.ReportData[0], PChar(AValue), iMaxCount);

end;



function TPublicVars.GetDestFile: string;

begin

  Result := FShareMem.DestFile;

end;



procedure TPublicVars.SetDestFile(AValue: string);

var

  iMaxCount: Integer;

begin

  ZeroMemory(@FShareMem.DestFile[0], Length(FShareMem.DestFile));

  iMaxCount := Min(Length(FShareMem.DestFile), Length(AValue));

  StrMove(@FShareMem.DestFile[0], PChar(AValue), iMaxCount);

end;



function TPublicVars.GetHintMsg: string;

begin

  Result := FShareMem.HintMsg;

end;



procedure TPublicVars.SetHintMsg(AValue: string);

var

  iMaxCount: Integer;

begin

  ZeroMemory(@FShareMem.HintMsg[0], Length(FShareMem.HintMsg));

  iMaxCount := Min(Length(FShareMem.HintMsg), Length(AValue));

  StrMove(@FShareMem.HintMsg[0], PChar(AValue), iMaxCount);

end;      



function TPublicVars.GetReportState: TReportState;

begin

  Result := FShareMem.ReportState;

end;



procedure TPublicVars.SetReportState(AValue: TReportState);

begin

  FShareMem.ReportState := AValue;

end;



end.




